An Integral of a Product of four Bessel Functions

Frédéric Chyzak

(Version of January 8, 1998)

In (Glasser, M. L. and Montaldi E. (1994): Some Integrals Involving Bessel Functions, *J. Math. Anal. Appl.*, **183**:577-590), Glasser and Montaldi compute a closed form for an integral of a product of two Bessel functions, and suggest that their treatment should extend to the following example

$$\int_0^\infty x\, J_1(a\,x)\, I_1(a\,x)\, Y_0(x)\, K_0(x)\, dx = -\frac{\ln(1-a^4)}{2\,\pi\,a^2},$$

which is of interest because it contains each of the four types of Bessel functions. This integral is one of numerous integrals containing four (or more) Bessel functions. See for instance (Prudnikov, A. P., Brychkov, Yu. A. and Marichev, O. I. (1986): *Integrals and Series. Volume 2: Special functions*, Gordon and Breach; Sec. 2.16.47).

In this session, we deal with the integral above and derive a closed form for it using our Mgfun package in an intimate interaction with the gfun package.

```
> with(Mgfun);
```

[*diag_of_sys*, *int_of_sys*, *pol_to_sys*, *sum_of_sys*, *sys*sys*, *sys+sys*]

```
> with(gfun);
```

[*Laplace*, *algebraicsubs*, *algeqtodiffeq*, *algeqtoseries*, *algfuntoalgeq*, *borel*, *cauchyproduct*, *diffeq*diffeq*, *diffeq+diffeq*, *diffeqtohomdiffeq*, *diffeqtorec*, *guesseqn*, *guessgf*, *hadamardproduct*, *holexprtodiffeq*, *invborel*, *listtoalgeq*, *listtodiffeq*, *listtohypergeom*, *listtolist*, *listtoratpoly*, *listtorec*, *listtoseries*, *listtoseries/Laplace*, *listtoseries/egf*, *listtoseries/lgdegf*, *listtoseries/lgdogf*, *listtoseries/ogf*, *listtoseries/revegf*, *listtoseries/revogf*, *maxdegcoeff*, *maxdegeqn*, *maxordereqn*, *mindegcoeff*, *mindegeqn*, *minordereqn*, *optionsgf*, *poltodiffeq*, *poltorec*, *ratpolytocoeff*, *rec*rec*, *rec+rec*, *rectodiffeq*, *rectohomrec*, *rectoproc*, *seriestoalgeq*, *seriestodiffeq*, *seriestohypergeom*, *seriestolist*, *seriestoratpoly*, *seriestorec*, *seriestoseries*]

More specifically, the *gfun* package will be used to prepare a system of PDE's for the application of *Mgfun* functions, and to solve the ODE that is output by the *Mgfun* package.

Search for a System of PDE's Satisfied by the Integrand $x\, J_1(a\,x)\, I_1(a\,x)\, Y_0(x)\, K_0(x)$

We use the gfun package to compute a system of PDE's satisfied by each factor of the integrand. Next, we use the Mgfun package to derive a system of PDE's satisfied by their product.

System of PDE's Satisfied by x

The identity function x trivially satisfies the following differential system

> $sys1 := \{x\left(\frac{\partial}{\partial x}\mathrm{h}(x,a)\right)-1, \frac{\partial}{\partial a}\mathrm{h}(x,a)\}$

where each entry *expr* in the set denotes the equation $expr = 0$.

System of PDE's Satisfied by $J_1(a\,x)$

We compute a system of PDE's for

> $f := \mathrm{BesselJ}(1, a\,x)$

by first computing an ODE with respect to the variable x using gfun[holexprtodiffeq], and next considering symmetries of f to derive a complete system.

```
> holexprtodiffeq(f,y(x));
```

$$\{(-1+a^2x^2)\,\mathrm{y}(x)+x\left(\frac{\partial}{\partial x}\mathrm{y}(x)\right)+x^2\left(\frac{\partial^2}{\partial x^2}\mathrm{y}(x)\right), \mathrm{y}(0)=0, \mathrm{D}(y)(0)=\frac{1}{2}a\}$$

```
> deq:=op(remove(type,",equation)):
```

> $\mathrm{subs}(\mathrm{y}(x)=\mathrm{h}(x,a), deq)$

$$(-1+a^2x^2)\,\mathrm{h}(x,a)+x\left(\frac{\partial}{\partial x}\mathrm{h}(x,a)\right)+x^2\left(\frac{\partial^2}{\partial x^2}\mathrm{h}(x,a)\right)$$

> $\mathrm{subs}(\{x=a, a=x, \mathrm{y}(x)=\mathrm{h}(x,a)\}, deq)$

$$(-1+a^2x^2)\,\mathrm{h}(x,a)+a\left(\frac{\partial}{\partial a}\mathrm{h}(x,a)\right)+a^2\left(\frac{\partial^2}{\partial a^2}\mathrm{h}(x,a)\right)$$

> $x\left(\frac{\partial}{\partial x}\mathrm{h}(x,a)\right)-a\left(\frac{\partial}{\partial a}\mathrm{h}(x,a)\right)$

```
> sys2:={""",""","}:
```

System of PDE's Satisfied by $I_1(a\,x)$

We compute a system of PDE's for

> $f := \mathrm{BesselI}(1, a\,x)$

in the same way as in the previous section.

```
> holexprtodiffeq(f,y(x));
```

$$\{(-1-a^2x^2)\,\mathrm{y}(x)+x\left(\frac{\partial}{\partial x}\mathrm{y}(x)\right)+x^2\left(\frac{\partial^2}{\partial x^2}\mathrm{y}(x)\right), \mathrm{y}(0)=0, \mathrm{D}(y)(0)=\frac{1}{2}a\}$$

```
> deq:=op(remove(type,",equation)):
```

> $\mathrm{subs}(\mathrm{y}(x)=\mathrm{h}(x,a), deq)$

$$(-1-a^2x^2)\,\mathrm{h}(x,a)+x\left(\frac{\partial}{\partial x}\mathrm{h}(x,a)\right)+x^2\left(\frac{\partial^2}{\partial x^2}\mathrm{h}(x,a)\right)$$

> $\mathrm{subs}(\{x=a, a=x, \mathrm{y}(x)=\mathrm{h}(x,a)\}, deq)$

$$(-1-a^2x^2)\,\mathrm{h}(x,a)+a\left(\frac{\partial}{\partial a}\mathrm{h}(x,a)\right)+a^2\left(\frac{\partial^2}{\partial a^2}\mathrm{h}(x,a)\right)$$

> $x\left(\frac{\partial}{\partial x}\mathrm{h}(x,a)\right)-a\left(\frac{\partial}{\partial a}\mathrm{h}(x,a)\right)$

```
> sys3:={""",""","}:
```

System of PDE's Satisfied by $Y_0(x)$

We compute a system of PDE's for

> $f := \mathrm{BesselY}(0, x)$

by first computing an ODE with respect to the variable x by using gfun[holexprtodiffeq], and next encoding that f does not depend on a.

```
> holexprtodiffeq(f,y(x));
```

$$x\left(\frac{\partial^2}{\partial x^2}\mathrm{y}(x)\right)+\left(\frac{\partial}{\partial x}\mathrm{y}(x)\right)+x\,\mathrm{y}(x)$$

> $sys4 := [\,\mathrm{subs}(\mathrm{y}(x)=\mathrm{h}(x,a), "), \frac{\partial}{\partial a}\mathrm{h}(x,a)\,]$

$$sys4 := [\frac{\partial}{\partial a}\mathrm{h}(x,a), x\left(\frac{\partial^2}{\partial x^2}\mathrm{h}(x,a)\right)+\left(\frac{\partial}{\partial x}\mathrm{h}(x,a)\right)+x\,\mathrm{h}(x,a)]$$

System of PDE's Satisfied by $K_0(x)$

We compute a system of PDE's for

> $f := \mathrm{BesselK}(0, x)$

in the same way as in the previous section.

```
> holexprtodiffeq(f,y(x));
```

$$x\left(\frac{\partial^2}{\partial x^2}\mathrm{y}(x)\right)+\left(\frac{\partial}{\partial x}\mathrm{y}(x)\right)-x\,\mathrm{y}(x)$$

> $sys5 := [\,\mathrm{subs}(\mathrm{y}(x)=\mathrm{h}(x,a), "), \frac{\partial}{\partial a}\mathrm{h}(x,a)\,]$

$$sys5 := [\frac{\partial}{\partial a}\mathrm{h}(x,a), x\left(\frac{\partial^2}{\partial x^2}\mathrm{h}(x,a)\right)+\left(\frac{\partial}{\partial x}\mathrm{h}(x,a)\right)-x\,\mathrm{h}(x,a)]$$

System of PDE's Satisfied by the Product

Computing a system for the product is now a simple call to Mgfun['sys*sys'].

```
> sys:=`sys*sys`(sys1,sys2,sys3,sys4,sys5);
```

$$\begin{aligned}sys := [\,&(-452\,a^4x^4+2091+740\,x^4)\,\mathrm{h}(x,a)+210\,a^2x^2\left(\frac{\partial^4}{\partial x^2\partial a^2}\mathrm{h}(x,a)\right)\\&+(-108\,x^3+700\,a^4x^7+20\,x^7)\left(\frac{\partial^3}{\partial x^3}\mathrm{h}(x,a)\right)-98\,x^5\left(\frac{\partial^5}{\partial x^5}\mathrm{h}(x,a)\right)\\&+140\,a\,x^4\left(\frac{\partial^5}{\partial x^4\partial a}\mathrm{h}(x,a)\right)+(140\,x^6+805\,x^2+420\,x^6a^4)\left(\frac{\partial^2}{\partial x^2}\mathrm{h}(x,a)\right)\\&+35\,x^6\left(\frac{\partial^6}{\partial x^6}\mathrm{h}(x,a)\right)+(-1908\,a\,x+2800\,a^5x^5+400\,a\,x^5)\left(\frac{\partial^2}{\partial x\,\partial a}\mathrm{h}(x,a)\right)\end{aligned}$$

$$- 770\, a\, x^3 \left(\frac{\partial^4}{\partial x^3\, \partial a} \mathrm{h}(x, a) \right) + 175\, x^4 \left(\frac{\partial^4}{\partial x^4} \mathrm{h}(x, a) \right)$$

$$+ (-1528\, a\, x^4 - 3272\, a^5\, x^4 + 1212\, a) \left(\frac{\partial}{\partial a} \mathrm{h}(x, a) \right)$$

$$+ (-412\, x^5 + 700\, a^4\, x^5 - 2091\, x) \left(\frac{\partial}{\partial x} \mathrm{h}(x, a) \right) + 5\, x^7 \left(\frac{\partial^7}{\partial x^7} \mathrm{h}(x, a) \right)$$

$$+ (-48\, a^3 - 400\, a^7\, x^4 + 400\, a^3\, x^4) \left(\frac{\partial^3}{\partial a^3} \mathrm{h}(x, a) \right)$$

$$+ (1514\, a\, x^2 - 1680\, a^5\, x^6 + 80\, a\, x^6) \left(\frac{\partial^3}{\partial x^2\, \partial a} \mathrm{h}(x, a) \right)$$

$$+ (1000\, x^4\, a^2 - 30\, a^2 - 2600\, x^4\, a^6) \left(\frac{\partial^2}{\partial a^2} \mathrm{h}(x, a) \right)$$

$$+ (-394\, a^2\, x + 200\, x^5\, a^2 + 1400\, a^6\, x^5) \left(\frac{\partial^3}{\partial x\, \partial a^2} \mathrm{h}(x, a) \right), (4\, a^4\, x^4 - 3 - 4\, x^4)\, \mathrm{h}(x, a)$$

$$- 6\, a^2\, x^2 \left(\frac{\partial^4}{\partial x^2\, \partial a^2} \mathrm{h}(x, a) \right) - x^2 \left(\frac{\partial^2}{\partial x^2} \mathrm{h}(x, a) \right) + 4\, a^3\, x \left(\frac{\partial^4}{\partial x\, \partial a^3} \mathrm{h}(x, a) \right)$$

$$+ 26\, a\, x \left(\frac{\partial^2}{\partial x\, \partial a} \mathrm{h}(x, a) \right) + 4\, a\, x^3 \left(\frac{\partial^4}{\partial x^3\, \partial a} \mathrm{h}(x, a) \right) - x^4 \left(\frac{\partial^4}{\partial x^4} \mathrm{h}(x, a) \right)$$

$$- 26\, a \left(\frac{\partial}{\partial a} \mathrm{h}(x, a) \right) + 3\, x \left(\frac{\partial}{\partial x} \mathrm{h}(x, a) \right) - 8\, a^3 \left(\frac{\partial^3}{\partial a^3} \mathrm{h}(x, a) \right) - 12\, a\, x^2 \left(\frac{\partial^3}{\partial x^2\, \partial a} \mathrm{h}(x, a) \right)$$

$$- 40\, a^2 \left(\frac{\partial^2}{\partial a^2} \mathrm{h}(x, a) \right) + 24\, a^2\, x \left(\frac{\partial^3}{\partial x\, \partial a^2} \mathrm{h}(x, a) \right), (-12\, a^4\, x^4 - 20\, x^4 + 21)\, \mathrm{h}(x, a)$$

$$+ 30\, a^2\, x^2 \left(\frac{\partial^4}{\partial x^2\, \partial a^2} \mathrm{h}(x, a) \right) - 3\, x^3 \left(\frac{\partial^3}{\partial x^3} \mathrm{h}(x, a) \right) - 3\, x^5 \left(\frac{\partial^5}{\partial x^5} \mathrm{h}(x, a) \right)$$

$$+ 10\, a\, x^4 \left(\frac{\partial^5}{\partial x^4\, \partial a} \mathrm{h}(x, a) \right) + 10\, x^2 \left(\frac{\partial^2}{\partial x^2} \mathrm{h}(x, a) \right) - 148\, a\, x \left(\frac{\partial^2}{\partial x\, \partial a} \mathrm{h}(x, a) \right)$$

$$- 20\, a\, x^3 \left(\frac{\partial^4}{\partial x^3\, \partial a} \mathrm{h}(x, a) \right) - 5\, x^4 \left(\frac{\partial^4}{\partial x^4} \mathrm{h}(x, a) \right)$$

$$+ (-8\, a\, x^4 + 8\, a^5\, x^4 + 172\, a) \left(\frac{\partial}{\partial a} \mathrm{h}(x, a) \right) + (-21\, x - 12\, x^5 - 20\, a^4\, x^5) \left(\frac{\partial}{\partial x} \mathrm{h}(x, a) \right)$$

$$- 8\, a^3 \left(\frac{\partial^3}{\partial a^3} \mathrm{h}(x, a) \right) + 64\, a\, x^2 \left(\frac{\partial^3}{\partial x^2\, \partial a} \mathrm{h}(x, a) \right) + 20\, a^2 \left(\frac{\partial^2}{\partial a^2} \mathrm{h}(x, a) \right)$$

$$
\begin{aligned}
&- 44\, a^2\, x \left(\frac{\partial^3}{\partial x\, \partial a^2} \mathrm{h}(x, a) \right) - 10\, x^3\, a^2 \left(\frac{\partial^5}{\partial x^3\, \partial a^2} \mathrm{h}(x, a) \right), (3 + 12\, a^4\, x^4 + 20\, x^4)\, \mathrm{h}(x, a) \\
&+ 16\, a^2\, x^2 \left(\frac{\partial^4}{\partial x^2\, \partial a^2} \mathrm{h}(x, a) \right) + 3\, x^3 \left(\frac{\partial^3}{\partial x^3} \mathrm{h}(x, a) \right) - 4\, x^5 \left(\frac{\partial^5}{\partial x^5} \mathrm{h}(x, a) \right) \\
&+ (-4\, x^6 - 2\, x^2 + 20\, x^6\, a^4) \left(\frac{\partial^2}{\partial x^2} \mathrm{h}(x, a) \right) - x^6 \left(\frac{\partial^6}{\partial x^6} \mathrm{h}(x, a) \right) \\
&+ (-8\, a\, x^5 - 190\, a\, x - 24\, a^5\, x^5) \left(\frac{\partial^2}{\partial x\, \partial a} \mathrm{h}(x, a) \right) - 20\, a\, x^3 \left(\frac{\partial^4}{\partial x^3\, \partial a} \mathrm{h}(x, a) \right) \\
&+ 4\, x^4 \left(\frac{\partial^4}{\partial x^4} \mathrm{h}(x, a) \right) + (-16\, a\, x^4 + 16\, a^5\, x^4 + 238\, a) \left(\frac{\partial}{\partial a} \mathrm{h}(x, a) \right) \\
&+ (-3\, x - 16\, x^5 + 16\, a^4\, x^5) \left(\frac{\partial}{\partial x} \mathrm{h}(x, a) \right) + 4\, a^3 \left(\frac{\partial^3}{\partial a^3} \mathrm{h}(x, a) \right) \\
&+ 76\, a\, x^2 \left(\frac{\partial^3}{\partial x^2\, \partial a} \mathrm{h}(x, a) \right) + (98\, a^2 + 8\, x^4\, a^6 - 8\, x^4\, a^2) \left(\frac{\partial^2}{\partial a^2} \mathrm{h}(x, a) \right) \\
&- 66\, a^2\, x \left(\frac{\partial^3}{\partial x\, \partial a^2} \mathrm{h}(x, a) \right) + 2\, a\, x^5 \left(\frac{\partial^6}{\partial x^5\, \partial a} \mathrm{h}(x, a) \right), -4\, a^3\, x^4\, \mathrm{h}(x, a) \\
&- a^3 \left(\frac{\partial^4}{\partial a^4} \mathrm{h}(x, a) \right) - 3 \left(\frac{\partial}{\partial a} \mathrm{h}(x, a) \right) - 4\, a^2 \left(\frac{\partial^3}{\partial a^3} \mathrm{h}(x, a) \right) + 3\, a \left(\frac{\partial^2}{\partial a^2} \mathrm{h}(x, a) \right) \}
\end{aligned}
$$

Note that the routine gfun['diffeq*diffeq'] performs a similar task, but is restricted to the univariate case, i.e., to functions described by a single ODE. Thus, it can only compute the ODE which does not involve cross-derivatives in the system above, and a similar equation with derivations with respect to x. It turns out that this would induce a loss of information, precluding the following calculation of the integral. Besides, Mgfun['sys*sys'] would also deal with systems of mixed differential-difference equations.

Integration

Again, the integration itself is performed by a single call to Mgfun[int_of_sys]:

```
> ode:=op(int_of_sys(sys,x=-infinity..infinity,takayama_algo));
```

$$
\begin{aligned}
ode := {}& 32\, a^3\, \mathrm{h}(a) + (a^7 - a^3) \left(\frac{\partial^4}{\partial a^4} \mathrm{h}(a) \right) + (-3 + 103\, a^4) \left(\frac{\partial}{\partial a} \mathrm{h}(a) \right) \\
&+ (-4\, a^2 + 16\, a^6) \left(\frac{\partial^3}{\partial a^3} \mathrm{h}(a) \right) + (73\, a^5 + 3\, a) \left(\frac{\partial^2}{\partial a^2} \mathrm{h}(a) \right)
\end{aligned}
$$

However, the justification that the algorithm selected by the option *takayama_algo* applies to the integral under consideration is rather technical and is beyond this presentation.

Resolution of the Final ODE

We use the gfun package to solve for the solution $\mathrm{h}(a)$ which corresponds to the integral to be computed. Due to its integral representation, this function is analytic at 0, hence admits a Taylor expansion at 0. We proceed to compute a closed form for $h(a)$ by summation of this expansion. To this end, we determine a recurrence equation on the coefficients of the Taylor expansion using gfun[diffeqtorec]:

> $ore := \mathrm{diffeqtorec}(ode, \mathrm{h}(a), \mathrm{u}(n))$

$$ore := [(n+2)\,\mathrm{u}(n) + (-n-6)\,\mathrm{u}(n+4),\ \mathrm{u}(1)=0,\ \mathrm{u}(3)=0]$$

Maple readily solves this recurrence:

```
> rsol:=rsolve(ore,u(n));
```

$$rsol := 2\,\frac{\left(\frac{1}{2}\mathrm{u}(2)+\frac{1}{4}\mathrm{u}(0)\right)\Gamma(n+2)}{\Gamma(n+3)} + 2\,\frac{\left(\frac{1}{4}\mathrm{u}(0)-\frac{1}{2}\mathrm{u}(2)\right)\mathrm{RootOf}(\_Z^2+1)^n\,\Gamma(n+2)}{\Gamma(n+3)}$$
$$+ 2\,\frac{\left(\frac{1}{2}\mathrm{u}(2)+\frac{1}{4}\mathrm{u}(0)\right)(-1)^n\,\Gamma(n+2)}{\Gamma(n+3)}$$
$$+ 2\,\frac{\left(\frac{1}{4}\mathrm{u}(0)-\frac{1}{2}\mathrm{u}(2)\right)(-\mathrm{RootOf}(\_Z^2+1))^n\,\Gamma(n+2)}{\Gamma(n+3)}$$

After rearranging the terms in the sum, it is obvious that $\mathrm{u}(n)$ is non-zero for even n only.

```
> collect(map(normal,rsol,expanded),u,factor);
```

$$\frac{1}{2}\,\frac{(1+\mathrm{RootOf}(\_Z^2+1)^n)(1+(-1)^n)\,\mathrm{u}(0)}{n+2} - \frac{(-1+\mathrm{RootOf}(\_Z^2+1)^n)(1+(-1)^n)\,\mathrm{u}(2)}{n+2}$$

We perform the corresponding change of variable, $n = 2\,p$,

```
> subs({n=2*p,(-1)^n=1,RootOf(_Z^2+1)^n=(-1)^p},");
```

$$\frac{(1+(-1)^p)\,\mathrm{u}(0)}{2\,p+2} - 2\,\frac{(-1+(-1)^p)\,\mathrm{u}(2)}{2\,p+2}$$

so that *Maple* can sum the Taylor series:

```
> sum("*a^(2*p),p=0..infinity);
```

$$\sum_{p=0}^{\infty}\left(\frac{(1+(-1)^p)\,\mathrm{u}(0)}{2\,p+2} - 2\,\frac{(-1+(-1)^p)\,\mathrm{u}(2)}{2\,p+2}\right)a^{(2\,p)}$$

```
> h:=collect(value(expand(")),u);
```

$$h := \left(-\frac{1}{2}\,\frac{\ln(-a^2+1)}{a^2} + \frac{1}{2}\,\frac{\ln(a^2+1)}{a^2}\right)\mathrm{u}(0) + \left(-\frac{\ln(-a^2+1)}{a^2} - \frac{\ln(a^2+1)}{a^2}\right)\mathrm{u}(2)$$

It only remains to evaluate u_0 and u_2. We first compute u_0 and find it is 0 by inversion of limits. Let

> $f := x\,\mathrm{BesselJ}(1, a\,x)\,\mathrm{BesselI}(1, a\,x)\,\mathrm{BesselY}(0, x)\,\mathrm{BesselK}(0, x)$

be the integrand. We have:

> $\displaystyle\int_0^{\infty} \lim_{a \to 0} f\,dx$

$$0$$

In the same way, each coefficient of the Taylor series for the integral is obtained by inversion of limits. In particular, $\kappa = u_2$, but *Maple* is not capable of integrating:

```
> kappa=int(coeff(series(normal(diff(f,a,a)),a=0),a,0)/2,x=0..infinity);
```

$$\kappa = \int_0^\infty \frac{1}{4} x^3 \operatorname{BesselY}(0, x) \operatorname{BesselK}(0, x)\, dx$$

(This integral for κ cannot be computed by a call to *int* using the Release 4, but the next release will probably be able to integrate it.)

We obtain the following form for $h(a)$:

```
> -combine(normal(-subs({u(0)=0,u(2)=kappa},h)),ln,symbolic);
```

$$-\frac{\ln((-a^2+1)(a^2+1))\,\kappa}{a^2}$$

It only remains to be proved that $\kappa = \frac{1}{2\pi}$. We do not do it, since computing this last integral which is a constant lies outside the scope of the theory of holonomy. With this example, we have reduced the problem of evaluating a parametrized integral to the evaluation of a non-parametrized integral. In case there were no closed form for κ, we could at least perform a simple numerical evaluation and return a result in terms of this numerical value and the series above for $\kappa = 1$.